Plot()
 {
  // d11b Removes boundaries from San Pedro dates
  // Adds the rest of the features from Big Tex to those first suggested by Erina
  // Identifies West house ring Feature 4124 as an outlier 9-28-2020
  //Treats all Cienega phase Strat 4 features as overlapping groups
  //Eliminates superpositioning in West house ring
  //Puts Strat 4 San Pedro house and pits contiguous between silt deposition and the Strat 4 pits and house rings
  //Inserts "Before Early Ceramic" TAQ
  // 5-11-2021 Updated with new colors and phase, boundary names, added sequential boundary between San Pedro and Early Cienega, demonstrates gap between them
  // 
  Sequence()
  {
   Boundary("Start Strat 500");
   //R_Combine based on common soil origin
   R_Combine("Strat 500 humic samples")
   {
    R_Date("AA:12:91 Big Tex-Humic F-0 Beta-563244", 3200, 30)
    {
    };
    R_Date("AA:12:91 Big Tex-Humic F-0 Beta-563253", 3170, 30)
    {
    };
    color="brown";
   };
   Boundary("Start Strat 500 Occupation");
   R_Date("AA:12:91 Big Tex F-4537 Beta-551692", 3070, 30)
   {
    color="#837465";
   };
   Interval("Strat 500 Occupation Interval");
   Boundary("End Strat 500 Occupation/Transition Massive silt");
   Date("Massive silt")
   {
    color="#eb984e";
   };
   Interval("Silt interval");
   Boundary("Transition Massive silt/Strat 4 San Pedro");
   //Features assigned to San Pedro phaae based on 14C date; all in Strat 4, cannot be distinguished from Cienega phase based on stratigraphy
   Phase("Strat 4 San Pedro phase")
   {
    Sequence("Strat 4 San Pedro phase South pits")
    {
     R_Date("AA:12:91 Big Tex F-4161 Beta-567287", 2800, 30)
     {
      color="blue";
     };
     R_Date("AA:12:91 Big Tex F-4162 Beta-567288", 2750, 30)
     {
      color="blue";
     };
    };
    Sequence("Strat 4 San Pedro phase South house")
    {
     R_Date("AA:12:91 Big Tex F-4616.01 Beta-567291", 2760, 30)
     {
      color="#5d6d7e";
     };
    };
   };
   Boundary("End San Pedro");
   Interval("Time Between San Pedro and Early Cienega");
   Boundary("Start Early Cienega");
   //Treats east house ring, west house ring, and pits as overlapping groups
   Phase("Strat 4 Early Cienega phase")
   {
    Sequence("Strat 4 pits")
    {
     Boundary("Start Strat 4 pits");
     Phase("Strat 4 pits")
     {
      KDE_Plot("Strat 4 Pits KDE", )
      {
      };
      R_Date("AA:12:91 Big Tex F-4520 Beta-551686", 2460, 30)
      {
       color="#800000";
      };
      R_Date("AA:12:91 Big Tex F-4541 Beta-551689", 2470, 30)
      {
       color="#800000";
      };
      R_Date("AA:12:91 Big Tex F-4551 Beta-551688", 2450, 30)
      {
       color="#800000";
      };
      R_Date("AA:12:91 Big Tex F-4561 Beta-551691", 2440, 30)
      {
       color="#800000";
      };
      R_Date("AA:12:91 Big Tex F-4192 Beta-567289", 2420, 30)
      {
       color="#800000";
      };
      R_Date("AA:12:91 Big Tex F-4263 Beta-567290", 2500, 30)
      {
       color="#800000";
      };
      R_Date("AA:12:91 Big Tex F-4809 Beta-567295", 2460, 30)
      {
       color="#800000";
      };
      R_Date("AA:12:91 Big Tex F-4114 Beta-567285", 2400, 30)
      {
       color="#800000";
      };
      Interval("Interval Strat 4 pits");
     };
     Boundary("End Strat 4 pits");
    };
    Sequence("Strat 4 East house ring")
    {
     Boundary("Start Strat 4 East house ring");
     Phase("Strat 4 East house ring")
     {
      KDE_Plot("Strat 4 East house ring KDE", )
      {
      };
      R_Date("AA:12:91 Big Tex F-4622.01 Beta-563248", 2430, 30)
      {
       color="#228B22";
      };
      R_Date("AA:12:91 Big Tex F-4644.02 Beta-563249", 2390, 30)
      {
       color="#228B22";
      };
      R_Date("AA:12:91 Big Tex F-4718.01 Beta-567292", 2440, 30)
      {
       color="#228B22";
      };
      Interval("Interval Strat 4 East house ring");
     };
     Boundary("End Strat 4 East house ring");
    };
    Sequence("Strat 4 West house ring")
    {
     Boundary("Start Strat 4 West house ring");
     Phase("Strat 4 West house ring")
     {
      KDE_Plot("Strat 4 West house ring KDE", )
      {
      };
      R_Date("AA:12:91 Big Tex F-4124 Beta-567286", 2490, 30)
      {
       color="#FF0000";
       Outlier()
       {
       };
      };
      R_Date("AA:12:91 Big Tex F-4890 Beta-567297", 2380, 30)
      {
       color="#008B8B";
      };
      R_Date("AA:12:91 Big Tex F-4849.02 Beta-48", 2420, 30)
      {
       color="#008B8B";
      };
      R_Date("AA:12:91 Big Tex F-4803.01 Beta-567294", 2390, 30)
      {
       color="#008B8B";
      };
      R_Date("AA:12:91 Big Tex F-4764 Beta-567293", 2380, 30)
      {
       color="#008B8B";
      };
      R_Date("AA:12:91 Big Tex F-4930.01 Beta-567298", 2350, 30)
      {
       color="#008B8B";
      };
      Interval("Interval Strat 4 West house ring");
     };
     Boundary("End Strat 4 West house ring");
    };
    Order("Order Early Cienega starts and ends")
    {
     Date("=Start Strat 4 pits");
     Date("=Start Strat 4 East house ring");
     Date("=Start Strat 4 West house ring");
     Date("=End Strat 4 pits");
     Date("=End Strat 4 East house ring");
     Date("=End Strat 4 West house ring");
    };
   };
   Boundary("Transition Strat 4 Early/Late Cienega");
   R_Date("AA:12:91 Big Tex F-4097.01 Beta-567283", 2100, 30)
   {
    color="#F4A460";
   };
   Boundary("End Strat 4 Late Cienega");
   Before("Before Early Ceramic", 120)
   {
   };
  };
 };
